# Maintainer: Maxime Gauduin <alucryd@archlinux.org>
# Maintainer: Tom Gundersen <teg@jklm.no>
# Contributor: Bartłomiej Piotrowski <bpiotrowski@archlinux.org>
# Contributor: tobias <tobias@archlinux.org>
# Contributor: Sarah Hay <sarah@archlinux.org>

pkgname=xvidcore
pkgver=1.3.5
pkgrel=1
pkgdesc='XviD is an open source MPEG-4 video codec'
arch=('x86_64')
url='http://www.xvid.org/'
license=('GPL')
depends=('glibc')
makedepends=('nasm')
provides=('libxvidcore.so')
source=("http://downloads.xvid.org/downloads/xvidcore-${pkgver}.tar.bz2")
sha256sums=('7c20f279f9d8e89042e85465d2bcb1b3130ceb1ecec33d5448c4589d78f010b4')

build() {
  cd xvidcore/build/generic

  ./configure \
    --prefix='/usr'
  make
}

package() {
  cd xvidcore/build/generic

  make DESTDIR="${pkgdir}" install
}

# vim: ts=2 sw=2 et:
